
Chapter 1.  Introduction
1.1  Welcome
Thank you for purchasing the PICDEM-1 demonstration board from Microchip 
Technology Incorporated. The PICDEM-1 is a simple board which 
demonstrates the capabilities of several Microchip microcontrollers, 
specifically from the PIC16C5X, PIC16C7XX, PIC16X8X, and PIC17C4X 
families.

The PICDEM-1 can be used stand-alone with a programmed part, or with an 
emulator system, such as PICMASTER®. Sample programs are provided to 
demonstrate the unique features of the supported devices.

The PICDEM-1 Kit comes with the following:

1. PICDEM-1 Demonstration Board
2. Sample device
3. Sample programs (3.5-inch disk)
4. PICDEM-1 Demonstration Board User’s Guide (this document)

If you are missing any part of the kit, please contact your nearest Microchip 
sales office, listed in the back of this publication, for help.

 

Figure 1.1:  PICDEM-1 Kit 

1.2  PICDEM-1 Demonstration Board
The PICDEM-1 demonstration board has the following hardware features:

1. 18-, 28- and 40-pin DIP sockets (Although 3 sockets are provided, only
one device may be used at a time.)

2. On-board +5V regulator for direct input from 9V, 100 mA AC/DC wall
adapter or hooks for a +5V, 100 mA regulated DC supply.

3. RS-232C socket and associated hardware for direct connection to
RS-232C interface.

4. 5K pot for devices with analog inputs.
5. Three push button switches for external stimulus and RESET.
6. Eight red LEDs connected to PORTB for displaying 8-bit binary values.
7. Socket for “canned” crystal oscillator.
8. Unpopulated holes provided for crystal connection.
9. Jumper to disconnect on-board RC oscillator (approximately 2 MHz).
10. Prototype area for user hardware.

Figure 1.2:  PICDEM-1 Hardware

1.3  Sample Devices
Several UV erasable devices are included. The device types may change. The 
supplied devices are typically one of the following:

• PIC16C54/55/56/57/58 or equivalent
• PIC16C71 or equivalent
• PIC16F83/84 or equivalent
• PIC17C42/43/44 or equivalent

1.4  Sample Programs
The PICDEM-1 Kit includes a 3.5” disk with sample demonstration programs 
on them. These programs may be used with the included sample device or 
with an emulator system. The programs are:

• tut.asm – Tutorial
• demo71.asm – PIC16C71 Demo
• demo84.asm – PIC16F84 Demo
• demo42.asm – PIC17C42 Demo

1.5  PICDEM-1 User’s Guide
This document describes the PICDEM-1 demonstration board, tutorial and 
demonstration software, to give the user a brief overview of the PICmicros® 
supported, as well as the PICMASTER. Detailed information on individual 
microcontrollers may be found in the device’s respective data sheet. Detailed 
information on the PICMASTER emulation system may be found in the 
PICMASTER User’s Guide (DS51037).

Chapter 1: Introduction – This chapter introduces the PICDEM-1 and 
provides a brief description of the hardware.

Chapter 2: Getting Started – This chapter goes through a basic step-by-step 
process for getting your PICDEM-1 up and running as a stand-alone board or 
on the PICMASTER emulator system.

Chapter 3: Tutorials – This chapter provides a detailed description of the 
tutorial programs.

Chapter 4: A/D Demo – PIC16C71 – This chapter provides a detailed 
description of the demonstration program for the A/D module on the 
PIC16C71.

Chapter 5: EEPROM Demo – PIC16F84 – This chapter provides a detailed 
description of the demonstration program for the data EEPROM on the 
PIC18F84.

Chapter 6: USART Demo – PIC17C42 – This chapter provides a detailed 
description of the demonstration program for the serial (RS-232) 
communications capabilities of the PIC17C42.

Appendix A: Hardware Detail: This appendix describes in detail the hardware 
of the PICDEM-1 board.

1.6  Reference Documents
Reference Documents may be obtained by contacting your nearest Microchip 
sales office (listed in the back of this document) or by downloading via the 
Microchip website (www.microchip.com).

• Microchip Technical Library CD-ROM (DS00161) or 
individual data sheets

• MPLABTM IDE, Simulator and Editor User’s Guide (DS51025)
• MPASM User’s Guide with MPLINK and MPLIB (DS33014)
• MPLAB PRO MATE User’s Guide (DS30082)
• PICSTART Plus User’s Guide (DS51028)
• MPLAB PICMASTER Emulator User’s Guide (DS30421)
• PICMASTER Probe Specification (DS51024)
• MPLAB-ICE User’s Guide (DS51159)
• Microchip Third Party Guide (DS00104)
• PICmicro Midrange Reference Manual (DS33023)
DS33015C-page 4  1999 Microchip Technology Inc.

Chapter 2.  Getting Started
The PICDEM-1 may be used as a stand-alone board or with an emulator. The 
emulator discussed in this chapter is the PICMASTER. However, other 
emulators may be used. For a list of PICmicro-compatible emulators, please 
refer to the Microchip Third Party Handbook or the Development Tool 
Ordering Guide.

2.1  PICDEM-1 as a Stand-alone Board – 
Preprogrammed Device

The PICDEM-1 may be demonstrated immediately by following the steps 
listed below:

• Make sure the preprogrammed sample device is in the appropriate 
socket on the PICDEM-1 board.

• Make sure there is a jumper on J3 (to enable the on-board RC oscilla-
tor).

• Apply power to the PICDEM-1 (Figure 2.1). For information on accept-
able power sources, see Appendix A.

• Press pushbutton S3 repeatedly to see the LEDs count up from 00h to 
FFh. Press pushbutton S2 to reset.

 

Figure 2.1:  PICDEM-1 Stand-Alone

2.2  PICDEM-1 as a Stand-alone Board – 
Sample Programs

To demonstrate PICDEM-1 operation with one of the sample programs, the 
sample device will have to be erased and reprogrammed. Once the device 
has been reprogrammed:

• Make sure the sample device is in the appropriate socket on the 
PICDEM-1 board.

• Make sure there is a jumper on J3 (to enable the on-board RC oscilla-
tor).

• Apply power to the PICDEM-1 (Figure 2.1). For information on accept-
able power sources, see Appendix A.

• Consult the appropriate chapter in this document for information on the 
execution of each demo.

2.2.1 Erasing the Sample Device
To erase an EPROM device:

• Remove any labels covering the device window. If you do not have a 
windowed device (Figure 2.2), you cannot reprogram it. A windowed 
version of all EPROM devices may be ordered by requesting the JW 
package.

• Place the device in an Ultraviolet (UV) EPROM Eraser. The amount of 
time required to completely erase a UV erasable device depends on: 
the wavelength of the light, its intensity, distance from UV source, and 
the process technology of the device (how small are the memory cells).

• Verify that the device is blank (e.g., perform a blank check) before 
attempting to program it.

Figure 2.2:  Windowed Device

To erase an EEPROM/Flash device:

• Enable your programmer in MPLAB. If you change programmers, you 
will have to restart MPLAB.

• Place the device in the programmer.

• Select Erase Program Memory or the equivalent erase command from 
the Programmer menu. 

2.2.2 Reprogramming the Sample Device
To reprogram the erased sample device, the following will be necessary:

1. Sample programs installed on the hard drive

The PICDEM-1 package includes a 3.5-inch disk which contains sample
programs for all the processor types supported. Instructions on how to
install the programs can be found in the readme file also on the disk.

2. An assembler, such as MPASM available with MPLAB

Sample programs may be used to program the sample device once they
have been assembled. Microchip Technology’s MPLAB Integrated Devel-
opment Environment (IDE) includes an assembler (MPASM). However,
other assemblers may be used. For a list of PICmicro-compatible assem-
blers, please refer to the Microchip Third Party Handbook.

3. A device programmer, such as PRO MATE® II or PICSTART® Plus

Once the sample program is in hex file format, a programmer may be used
to program a blank device. Microchip Technology’s PRO MATE II or
PICSTART Plus programmers may be used. Both are compatible with
MPLAB. However, other programmers may be used. For a list of PICmicro-
compatible programmers, please refer to the Microchip Third Party Hand-
book.

If the code protection bit(s) have not been programmed, the on-chip program 
memory can be read out for verification purposes. 

Note: You do not have to erase an EEPROM/Flash device before repro-
gramming it.

Note: Microchip does not recommend code protecting windowed
devices.
 1999 Microchip Technology Inc. DS33015C-page 7

2.3  PICDEM-1 Used with the PICMASTER Emulator – 
Sample Programs

To use the PICDEM-1 with the PICMASTER emulator, the following will be 
necessary:

1. PICDEM-1 demonstration board.
2. PICMASTER emulator pod.
3. PICMASTER device-specific probe.
4. Logic probes with power supply pins or external power supply (optional).
5. PC with MPLAB IDE software and MPLAB User’s Guide.
6. Ribbon cable to connect pod with probe.
7. Pin connector to connect probe with PICDEM-1.

 

Figure 2.3:  PICDEM-1 with the PICMASTER Emulator

This chapter goes through a basic step-by-step process for getting your 
PICDEM-1 up and running using the PICMASTER emulator system.

1. Assemble the PICMASTER emulator

• Follow the instructions in your PICMASTER emulator package and con-
nect the emulator pod to an IBM® compatible PC. Make certain that 
the power switches for both the PC and the PICMASTER are in the 
OFF position before connecting.

• Connect the appropriate probe to the PICMASTER pod using the rib-
bon cable provided (6 of Figure 2.3). For information on which probe to 
choose, refer to the PICMASTER Probe Specification.

• Connect the logic probes to the emulator pod.

2. Prepare the PICDEM-1

• Remove any preprogrammed microcontrollers installed in the PIC-
DEM-1.

• Make sure there is a jumper on J3 (to enable the on-board RC oscilla-
tor).

3. Prepare the probe – AC1650011 Example

• Determine the power source and set J5 accordingly; +5 VSYS comes 
from the PICMASTER system (internal) or +5 VEXT comes from an 
external power source.

• Determine whether or not to enable master clear MCLR by connecting/
disconnecting J6. It is recommended that J6 be connected (MCLR 
enabled) for most applications.

• Select the oscillator option to be used. Table 2.1 lists the modifications 
necessary for each option.

TABLE:  2.4  OSCILLATOR SELECTION

Oscillator 
selection on 
PICDEM-1

Modification on PICDEM-1 Modification on Probe

RC J3 installed, Y2 empty, 
Y1 empty

Select J4 = EXTCLK,
Select SW1 = 11 (RC)

Crystal J3 removed, Y2 empty, 
crystal in Y1, caps C3 and 
C4 installed

Select J4 = EXTCLK,
Select SW1 = 10 (HS), 
01 (XT) or 00 (LP)

Canned 
Crystal

J3 removed, oscillator in Y2, 
Y1, C3, C4 empty

Select J4 = EXTCLK,
Select SW1 = 10 (HS), 
01 (XT) or 00 (LP)

PICMASTER’s 
Oscillator

J3 removed, Y2, Y1 empty Select J4 = INTCLK

4. Connect the PICDEM-1 to the emulator.

• Connect the PICMASTER header probe to the appropriate socket on 
the PICDEM-1 using a connector (7 of Figure 2.3). Example: If you are 
emulating the PIC16F84, connect the 18-pin header on the probe card 
to the 18-pin socket on PICDEM-1 (U2).

5. Power supply considerations.

• If the PICMASTER power probes are to be used to power the PICDEM-
1, connect the Red probe clip to the hook marked +5V on PICDEM-1 
and the Black probe clip to the hook marked GND on PICDEM-1. 

6. Power-up sequence

The following power up sequence must be followed exactly:

• Power up the PC.
• Start Microsoft® Windows®.
• Power up the PICMASTER.
• If an external wall mount power supply is being used for the PICDEM-1, 

connect it now.
7. Installing the sample programs

The PICDEM-1 package includes a 3.5-inch disk which contains sample
programs for all the processor types supported. Instructions on how to
install the programs can be found in the readme file also on the disk.

8. Starting MPLAB

• Start MPLAB by double clicking on the MPLAB icon (Windows 3.1 or 
later) or selecting it from the Start menu (Windows 95/98).

• Refer to the MPLAB User’s Guide for how to configure the application 
to work with the emulator and how to run the demonstration programs.

2.4  PICDEM-1 Used with MPLAB-ICE – Sample 
Programs

Microchip’s latest emulator, MPLAB-ICE, may be used with the PICDEM-1 
board by using PICMASTER probes. Refer to the MPLAB-ICE User’s Guide 
(DS51159) for information on how to use PICMASTER probes with 
MPLAB-ICE. Then follow the setup directions from the previous section.

Note: If an external wall mount power supply is being used, do not con-
nect it to the PICDEM-1 yet.
DS33015C-page 10  1999 Microchip Technology Inc.

Chapter 3.  Tutorials
The tut.asm tutorial program is pre-programmed into the sample device. 
This program is listed on the included 3.5-inch program disk for user 
reference. For example, if the sample device has been reprogrammed with 
another sample program, the tutorial may be reassembled and reprogrammed 
into the device.

The tutorial program functions as follows. Pressing the switch S3 causes the 
LEDs to binary count up to FFh (255 decimal). Past FFh (all LEDs on), the 
count rolls over to zero (all LEDs off). The program may be reset by pressing 
the switch S2.

For detailed information on the PICDEM-1 hardware, please refer to 
Appendix A.

 

Figure 3.1:  Tutorial

3.1  Main Routine
The tutorial program is extremely simple. It begins by configuring PORTB as 
all outputs to drive the eight LEDs and initializing a counter. Then it waits in an 
infinite loop until S3 is pressed. This increases the counter by one, and this 
number is displayed in binary on the LEDs. Once the count reaches FFh, the 
next S3 press will roll the count back to 00h.

Figure 3.2:  Main Routine

3.2  Source Code – PIC16C54
The tutorial program listed below is for the PIC16C54 device. It can be 
modified for all other supported devices. See files titled tut??.asm, where 
?? represents the last two numbers of a device.

             list p=16c54
;
;  This program runs on the PICDEM-1 demo board.
;  In the Demo board, Port B is connected to 8 LEDs. 
;  RA1 is connected to a switch (S3). This program increments
;  the file register count each time S3 is pressed.
;  The value of count is displayed on the LEDs connected
;  to Port B.
;  Net result is that Leds should increment in a binary
;  manner every time S3 is pressed.
;
     #include <P16C5X.INC>
;
COUNT   equ     0x10

;
     org     00h
Start
     movlw   0
     movwf   PORTB           ; config port b as output
     tris    PORTB
     clrf    COUNT           ; clr count
Loop
     btfss   PORTA,1         ; see if RA1 pressed
     goto    IncCount        ; yes then inc count
Endloop
     goto    Loop            ; else check again
IncCount
     incf    COUNT,F         ;inc count
     movf    COUNT,W         ;
     movwf   PORTB           ; display on port b
Debounce
     btfss   PORTA,1         ; wait for key release
     goto    Debounce        ; not release then wait
Enddebounce
     goto    Loop            ; else check key press again
;
;
     org     0x01FF           ; 16C54 reset vector.
     goto    Start
;
     end
 1999 Microchip Technology Inc. DS33015C-page 13

NOTES:

Chapter 4.  A/D Demo – PIC16C71
The demo71.asm program is a simple implementation of the PIC16C71’s 
analog-to-digital (A/D) converter. The program reads A/D channel 0 and 
displays the results on the LEDs connected to PORTB. If the potentiometer is 
turned all the way clockwise, all of the LEDs are off. If the potentiometer is 
turned all the way counter-clockwise, all of the LEDs are turned on.

PIC16C5X and PIC16X8X devices do not have an on-board A/D converter. 
However, an external A/D circuit may be constructed in the prototype area and 
the board modified to accommodate it. For more information, please refer to 
AN513 – Analog to Digital Conversion Using a PIC16C54 (DS00513).

For more information on A/D module operation, please refer to the PICmicro 
Midrange Microcontroller Family Reference Manual for an operational 
description and a list of related application notes. For detailed information on 
the PICDEM-1 hardware, please refer to Appendix A.

 

Figure 4.1:  A/D Demo

4.1  Main Routine
The main routine of demo71.asm handles the initialization of the PIC16C71 
and then uses the A/D interrupt flag to signal a completed A/D conversion and 
update the value on the LEDs. The first few lines set PORTB as outputs (for 
the LEDs). InitializeAD, the Analog-to-Digital converter initialization 
routine, is then called. After the return from InitializeAD, the value in the 
A/D output register is retrieved and displayed on the LEDs. The routine then 
waits for TAD (A/D conversion clock) and then starts an A/D conversion of a 
value specified by the potentiometer. When the A/D conversion is complete 
(ADIF bit set), the program loops back to display the current conversion value 
on the LEDs and to get another A/D value for conversion.

Figure 4.2:  Main Routine

4.2  InitializeAD Routine
The InitializeAD routine sets up the A/D converter for use with the 
PICDEM-1 board. The routine sets Ch0 through Ch3 as analog inputs, sets 
the A/D to use the internal RC oscillator, selects channel 0, and clears the A/D 
result register.

 

Figure 4.3:  Initialize A/D

4.4  Detailed Code Description
The program starts out by defining three constants: TEMP, a general purpose 
file register; ADIF, the bit location (1) of the A/D interrupt flag in ADCON0; and 
ADGO, the A/D GO/DONE bit in ADCON0. The next sections of code define 
the reset and interrupt vectors.

TEMP    equ     10h
ADIF    equ     1               ;A/D interrupt flag
ADGO    equ     2               ;A/D enable bit
;
     org     0x00
     goto    Start              ;reset vector
;
     org     0x04
     goto    Service_int        ;interrupt vector

The actual program begins at Start, where PORTB is defined as all outputs by 
writing 0’s to the PORTB tristate register TRISB. Note that 0’s were written to 
the PORTB register prior to setting the port to outputs so that the port would 
initialize with all outputs in the low state.

Start
     movlw   B’00000000’     ;set port b as
     movwf   PORTB           ;all outputs
     bsf     STATUS, RP0
     CLRF    TRISB           ;       /
     bcf     STATUS, RP0
;
     call    InitializeAD

After initializing PORTB, InitializeAD  is called. The page register is set for 
page 1, RA0 through RA3 are selected to be analog inputs, and VREF is 
selected to be VDD (if all four pins are selected to be analog inputs, VREF is 
set to VDD by default) by writing to ADCON1. The page register is then reset 
for page 0, the internal RC oscillator is selected for the conversion clock, and 
ADON, the A/D on/off bit is set so that the A/D converter module is operating. 
The last action this routine takes before returning is to clear the A/D result 
register.

InitializeAD
     bsf     STATUS,5        ;select pg1
     movlw   B’00000000’     ;select ch0-ch3...
     movwf   ADCON1          ;as analog inputs
     bcf     STATUS,5        ;select pg0
     movlw   B’11000001’     ;select:RC,ch0..
     movwf   ADCON0          ;turn on A/D.
     clrf    ADRES           ;clr result reg.
     return

Chapter 4. A/D Demo – PIC16C71
The Update routine reads the A/D result register, writes the value it got to the 
LEDs connected to PORTB, calls SetupDelay, clears the interrupt flag and 
enables the next conversion. 

Update
     movf    ADRES,W         ;get a/d value
     movwf   PORTB           ;output to port b
     call    SetupDelay      ;setup time >= 20uS.
     bcf     ADCON0,ADIF     ;clear int flag
     bsf     ADCON0,ADGO     ;start new conversion

The SetupDelay routine provides a software delay so that the sample and 
hold circuit has enough time to settle before the A/D conversion is started. 
The SD loop takes 2 µs when TEMP is not equal to zero, so a value of nine 
loaded into TEMP would provide a delay of 18 µs. With the 2 µs execution 
time of the two moves before the SD loop and the call and return execution 
times, a delay of greater than 20 µs is provided for settling.

SetupDelay
     movlw   .3
     movwf   TEMP
SD
     decfsz  TEMP, F
     goto    SD
     return

The Loop routine tests the A/D interrupt flag, and loops until it indicates that 
the conversion has finished (ADIF = 0). When the conversion has finished, it 
jumps back to Loop so that the A/D value can be read and output to the LEDs.

Loop
     btfsc   ADCON0,ADIF     ;a/d done?
     goto    Update          ;yes then update new value.
     goto    Loop            ;no then keep checking

4.5  Source Code – PIC16C71
             LIST P=16C71
;
;  TITLE   "Single channel A/D (SAD)"
;  This program is a simple implementation of the PIC16C71’s
;  A/D. One channel is used (CH0).
;  The A/D is configured as follows:
;       Vref = +5V internal.
;       A/D Osc. =  internal RC
;       A/D Channel = CH0
;  Hardware for this program is the PICDEM1 board. The program 
;  converts the potentiometer value on CH0 and displays it as
;  a 8 bit binary value on PORTB.
;
;
       #include <P16C71.INC>
;
TEMP    equ     10h
ADIF    equ     1               ;A/D interrupt flag
ADGO    equ     2               ;A/D enable bit
;
     org     0x00
     goto    Start           ;reset vector
;
     org     0x04
     goto    Service_int     ;interrupt vector
;
;
     org     0x10
Start
     movlw   B’00000000’     ;set port b as
     movwf   PORTB           ;all outputs
     bsf     STATUS, RP0
     CLRF    TRISB           ;       /
     bcf     STATUS, RP0
;
     call    InitializeAD
Update
     movf    ADRES,W         ;get a/d value
     movwf   PORTB           ;output to port b
     call    SetupDelay      ;setup time >= 20uS.
     bcf     ADCON0,ADIF     ;clear int flag
     bsf     ADCON0,ADGO     ;start new conversion
Loop
     btfsc   ADCON0,ADIF     ;a/d done?
     goto    Update          ;yes then update new value.
     goto    Loop            ;no then keep checking
;

;No interrupts are enabled, so if the program ever reaches here,
;it will be returned with the global interrupts disabled.

Service_int
     return                  ;do not enable global.
;
;
;
;Initializes and sets up the A/D hardware.
;Select ch0 to ch3 as analog inputs, fosc/2 and read ch0.
;
InitializeAD
     bsf     STATUS,5        ;select pg1
     movlw   B’00000000’     ;select ch0-ch3...
     movwf   ADCON1          ;as analog inputs
     bcf     STATUS,5        ;select pg0
     movlw   B’11000001’     ;select:RC,ch0..
     movwf   ADCON0          ;turn on A/D.
     clrf    ADRES           ;clr result reg.
     return
;
;This routine is a software delay of 20uS for the a/d setup.
;At 4Mhz clock, the loop takes 2uS, so initialize TEMP with
;a value of 9 to give 18uS, plus the move etc. should result in
;a total time of > 20uS.

SetupDelay
     movlw   .3
     movwf   TEMP
SD
     decfsz  TEMP, F
     goto    SD
     return

     END

NOTES:

Chapter 5.  EEPROM Demo – PIC16F84
The demo84.asm program demonstrates the use of the on-board data 
EEPROM of the PIC16X8X devices. The program writes 00h through FFh to 
each address of the data EEPROM, and then reads back the value. The LEDs 
indicate the data presently being written. If an error occurs, all of the LEDs 
flash.

For detailed information on the PICDEM-1 hardware, please refer to 
Appendix A.

 

Figure 5.1:  EEPROM Demo

5.1  Main Routine
The program begins by clearing all interrupt flags and configuring PORTB as 
outputs. Then the value of PORTB is used, plus one, as the value written to all 
EEPROM locations (WriteAll). These values are then checked 
(CheckAll). If no errors are found, the value displayed on PORTB is 
increased by one and the program loops back to use this value, plus one, to 
write to all EEPROM locations again. Once the value written has reached 
FFh, adding one will cause it to roll over to 00h.

If an error is encountered during the check, the program branches to 
BlinkLEDs, which flashes the LEDs.

Figure 5.2:  Main Routine

5.2  WriteAll Routine
This routine writes the value on the LEDs, plus one, to all EEPROM locations.

 

Figure 5.3:  Write All

5.3  CheckAll Routine
This routine checks all of the EEPROM locations written by WriteAll. If an 
error is encountered, it branches to the BlinkLEDs routine.

Figure 5.4:  Check All

5.4  BlinkLEDs Routine
This routine flashes the LEDs if an error is encountered while checking the 
values written to the EEPROM.

Figure 5.5:  BlinkLEDs

5.5  Detailed Code Description

the flashing of the LEDs, GPFLAG, a general purpose flag register, and 

TIME      equ     0x10

EEPERROR  equ     0               ;GPFLAG,0

The next section of code performs the initialization of GPFLAG and PORTB.

Start
     clrf    GPFLAG          ;clr all flags
     movlw   B’00000000’     ;make port b outputs
     movwf   PORTB
     bsf     STATUS, RP0
     clrf    TRISB           ;       /
     bcf     STATUS, RP0

After the initialization is completed the program goes into a loop where all 
locations of the data EEPROM are written to and then read from for 
verification. If all locations verify correctly, the value displayed on the LEDs is 
incremented, if a mismatch between what was written and what is read is 
found, the LEDs are blinked at a one Hertz rate.

Next
     call    WriteAll        ;write to all locations
     call    CheckAll        ;verify all locations
     btfsc   GPFLAG,EEPERROR ;no error then skip
     goto    BlinkLeds       ;else blink leds.
     incf    PORTB, F        ;inc value in port b
     goto    Next

WriteAll  writes the value of what is displayed on the LEDs plus one to each 
location of the EEPROM. Notice that the write enable bit, WREN, in the 
EEPROM control register, EECON1 is set for the entire duration and does not 
have to be re-enabled for each byte.

WriteAll
     clrf    EEADR           ;start at addr = 0
     incf    PORTB,W         ;read current value+1
     movwf   EEDATA          ;ld. data reg.
     bsf     STATUS,RP0      ;select pg 1
     bsf     EECON1,WREN     ;enable write operation
     bcf     STATUS,RP0      ;select pg 0
WA1
     call    WriteOne        ;write to a location
     incf    EEADR, F        ;inc address
     btfss   EEADR,6         ;all 64 done?
     goto    WA1             ;no then do next
     bsf     STATUS,RP0      ;pg 1
     bcf     EECON1,WREN     ;disable write
     bcf     STATUS,RP0      ;pg 0
     return

WriteOne is where the actual writing of the EEPROM takes place. EEPROM 
writes will not initiate if the proper sequence (write 55h to EECON2, write AAh 
to EECON2, then set the WR bit) is not followed with exact timing. In order to 
ensure this, interrupts have not been enabled. After the writing of the data, 
and initiating the write sequence, the loop, W01, waits for the WR bit to go low 
indicating that the write has completed. 

WriteOne
     bsf     STATUS,RP0      ;page 1
     movlw   0x55            ;do write seq.
     movwf   EECON2          ;       /
     movlw   0xaa            ;      / 
     movwf   EECON2          ;     /  
     bsf     EECON1,WR       ;initiate write
WO1     
     btfsc   EECON1,WR       ;write complete?
     goto    WO1             ;no then keep checking
     bcf     STATUS,RP0      ;pg 0
     return

Refer to PIC16F8X Data Sheet for EEPROM programming requirements.

CheckAll reads each location of the EEPROM and verifies that it matches 
the value displayed on the LEDs plus one. If the value does not match, bit 
EEPERROR of file register GPFLAG will be set and the routine will return 
without verifying the other locations.

CheckAll
     clrf    EEADR           ;start at addr = 0
     incf    PORTB,W         ;ld. data to inspect
CA1
     call    CheckOne        ;check location
     btfsc   GPFLAG,EEPERROR ;any error?
     return                  ;yes then quit
     incf    EEADR, F        ;inc address
     btfss   EEADR,6         ;all 64 checked?
     goto    CA1             ;no then do next
     return

CheckOne does the actual reading of the EEPROM byte and verifying that it 
is correct. Although it is not necessary to check any flag bits during EEPROM 
reads, they can be done immediately after setting the address in EEADR and 
setting the read control bit (RD EECON1<0>), the RD bit will be cleared by the 
hardware when the data has been placed in the EEDATA register. If there is a 
verification error, the EEPERROR bit is set in the GPFLAG register.

Note: The timing in WriteOne is critical. Do not single step or do a soft-
ware trace through this routine.

CheckOne
     bsf     STATUS,RP0      ;pg 1
     bsf     EECON1,RD       ;do a read
CO1
     btfsc   EECON1,RD       ;rd done?
     goto    CO1             ;no then loop
     bcf     STATUS,RP0      ;pg 0
     xorwf   EEDATA, F       ;compare?
     btfss   STATUS,Z        ;same then skip
     bsf     GPFLAG,EEPERROR ;set error flag
     return

BlinkLeds sets up the RTCC to interrupt upon time-out, turns off the LEDs, 
and when one second has expired, toggles the LEDs.

BlinkLeds
     call    InitRtcc
     clrf    PORTB           ;turn off leds
BL1
     call    SecondOver      ;wait for 1/2 second
     comf    PORTB, F        ;toggle leds
     goto    BL1

InitRtcc assigns the prescaler to the RTCC and sets it for divide by 256. It 
then enables interrupts and initializes the TIME register to eight.

InitRtcc
     bsf     STATUS, RP0
     movlw   B’10000111’     ;rtcc inc = tcylc/256
     movwf   OPTION_REG      ;       /
     bcf     STATUS, RP0
     clrf    TMR0            ;start time
     movlw   B’10100000’     ;enable interrupts
     movwf   INTCON          ;       /
     movlw   .8              ;initialize time
     movwf   TIME
     return

ServiceRtcc  is the interrupt service routine and will be called each time the 
RTCC times out. It first verifies that the interrupt was caused by the RTCC and 
if it was it decrements TIME. Whether it was an RTCC interrupt or not, all 
interrupts are cleared and the RTCC interrupt is enabled before returning.

ServiceRtcc
     btfsc   INTCON,T0IF     ;rtcc interrupt?
     decf    TIME, F         ;yes then dec time
     clrf    INTCON          ;clr all interrupts
     bsf     INTCON,T0IE     ;enable RTIE
     retfie                  ;not zero then return

SecondOver waits for the TIME register to reach zero, restores a count of 
eight in it, and returns.

SecondOver
     movf    TIME,W          ;check if time = 0
     btfss   STATUS,Z        ;       /
     goto    SecondOver      ;no then loop
     movlw   .8              ;load for 1/2 second
     movwf   TIME
     return

5.6  Source Code – PIC16F84
     list p=16f84
;
;  This program demonstrates the PIC16F84.
;  To demo the EEPROM capability of the PIC16F84, then
;  a value is incrementally written and verified in all
;  64 locations of the EEPROM. If an error occurs at any location,
;  then all leds on portb will blink at a 1 sec rate.
;
     #include <P16F84.INC>
;
TIME      equ     0x10
GPFLAG    equ     0x20            ;Define flag register
EEPERROR  equ     0               ;GPFLAG,0
;
     org     0
     goto    Start
;
     org     4
     goto    ServiceRtcc
;
     org     10
Start
     clrf    GPFLAG          ;clr all flags
     movlw   B’00000000’     ;make port b outputs
     movwf   PORTB
     bsf     STATUS, RP0
     clrf    TRISB           ;       /
     bcf     STATUS, RP0
Next
     call    WriteAll        ;write to all locations
     call    CheckAll        ;verify all locations
     btfsc   GPFLAG,EEPERROR ;no error then skip
     goto    BlinkLeds       ;else blink leds.
     incf    PORTB, F        ;inc value in port b
     goto    Next
;
WriteAll
     clrf    EEADR           ;start at addr = 0
     incf    PORTB,W         ;read current value+1
     movwf   EEDATA          ;ld. data reg.
     bsf     STATUS,RP0      ;select pg 1
     bsf     EECON1,WREN     ;enable write operation
     bcf     STATUS,RP0      ;select pg 0

WA1
     call    WriteOne        ;write to a location
     incf    EEADR, F        ;inc address
     btfss   EEADR,6         ;all 64 done?
     goto    WA1             ;no then do next
     bsf     STATUS,RP0      ;pg 1
     bcf     EECON1,WREN     ;disable write
     bcf     STATUS,RP0      ;pg 0
     return
;
WriteOne
     bsf     STATUS,RP0      ;page 1
     movlw   0x55            ;do write seq.
     movwf   EECON2          ;       /
     movlw   0xaa            ;      / 
     movwf   EECON2          ;     /  
     bsf     EECON1,WR       ;initiate write
WO1     
     btfsc   EECON1,WR       ;write complete?
     goto    WO1             ;no then keep checking
     bcf     STATUS,RP0      ;pg 0
     return
;
CheckAll
     clrf    EEADR           ;start at addr = 0
     incf    PORTB,W         ;ld. data to inspect
CA1
     call    CheckOne        ;check location
     btfsc   GPFLAG,EEPERROR ;any error?
     return                  ;yes then quit
     incf    EEADR, F        ;inc address
     btfss   EEADR,6         ;all 64 checked?
     goto    CA1             ;no then do next
     return
;
CheckOne
     bsf     STATUS,RP0      ;pg 1
     bsf     EECON1,RD       ;do a read
CO1
     btfsc   EECON1,RD       ;rd done?
     goto    CO1             ;no then loop
     bcf     STATUS,RP0      ;pg 0
     xorwf   EEDATA, F       ;compare?
     btfss   STATUS,Z        ;same then skip
     bsf     GPFLAG,EEPERROR ;set error flag
     return
;

BlinkLeds
     call    InitRtcc
     clrf    PORTB           ;turn off leds
BL1
     call    SecondOver      ;wait for 1/2 second
     comf    PORTB, F        ;toggle leds
     goto    BL1
;
InitRtcc
     bsf     STATUS, RP0
     movlw   B’10000111’     ;rtcc inc = tcylc/256
     movwf   OPTION_REG      ;       /
     bcf     STATUS, RP0
     clrf    TMR0            ;start time
     movlw   B’10100000’     ;enable interrupts
     movwf   INTCON          ;       /
     movlw   .8              ;initialize time
     movwf   TIME
     return
;
ServiceRtcc
     btfsc   INTCON,T0IF     ;rtcc interrupt?
     decf    TIME, F         ;yes then dec time
     clrf    INTCON          ;clr all interrupts
     bsf     INTCON,T0IE     ;enable RTIE
     retfie                  ;not zero then return
;
SecondOver
     movf    TIME,W          ;check if time = 0
     btfss   STATUS,Z        ;       /
     goto    SecondOver      ;no then loop
     movlw   .8              ;load for 1/2 second
     movwf   TIME
     return                  

     end

Chapter 6.  USART Demo – PIC17C42
The demo42.asm program demonstrates the USART peripheral of the 
PIC17C42. The program waits for S1 to be pressed, increments a counter, 
and transmits it through the serial port. The RX and TX lines of the serial port 
must be shorted at the DB9 connector (pins 2 and 3). The value received from 
the serial port is displayed on the LEDs.

Other devices do not have a serial port implemented in hardware. However, if 
the user wishes to use a serial port, one may be implemented in software. For 
more information, please refer to AN555 – Software Implementation of 
Asynchronous Serial I/O (DS00555).

For more information on USART module operation, please refer to the 
PICmicro Midrange Microcontroller Family Reference Manual for an 
operational description and a list of related application notes. For detailed 
information on the PICDEM-1 hardware, please refer to Appendix A.

Figure 6.1:  USART Demo

6.1  Main Routine
The program demo42.asm begins by setting up the interrupts. The main 
routine actually commences at Start. It first clears data memory (RAM), and 
then calls two subroutines to initialize the ports. After initialization, the 
program waits for S1 to be pressed, then goes to the Xmit routine. On 
returning from transmit, the program loops to wait for another key press.

Figure 6.2:  Main Routine

6.2  Service_pref and Service_recv Routines
The routines Service_pref and Service_recv are the peripheral 
interrupt service routines. When an interrupt occurs, the program goes to 
Service_pref. Interrupts are disabled, and the receive buffer is checked. If 
it is not full, interrupts are re-enabled and routine returns to the main program. 
If the buffer is full, then Service_recv is called. This routine outputs the 
buffer data to the LEDs, clears the interrupts and then re-enables the receive 
interrupt.

Figure 6.3:  Service_pref Routine

Figure 6.4:  Service_recv Routine

6.3  Init_ports and Init_serial Routines
The Init_ports routine simply initializes PORTB.

Figure 6.5:  Initialize Ports Routine

Init_serial initializes the USART. The baud rate is set to 9600, reception 
and transmit are enabled, and corresponding interrupts are also enabled.

Figure 6.6:  Initialize Serial Port (USART)

6.4  Xmit Routine
The Xmit routine increases the value of count by one and then transmits this 
value. Then the routine waits for S1 to be released (debounce) before 
returning to the main routine.

Figure 6.7:  Transmit Routine

Start                           ; start main program
        movlw   0x20            ; clear ram space
        movfp   WREG,FSR0       ; use indirect addressing
Start1
        clrf    INDF0, F        ; clear ram
        incfsz  FSR0, F         ; from 0x20 to 0xff
        goto    Start1          ;       /
        call    Init_ports      ; initialize all ports
        call    Init_serial     ; initialize serial port
Loop    
        movpf   PORTA,TEMP      ; get porta
        btfss   TEMP,1          ; check for key press
        call    Xmit            ; send next value
        goto    Loop            ; check again

The routine Init_ports sets all pins of PORTB to be outputs and sets them 
to the low state (LEDs off).

Init_ports
        movlb   0               ; select bank 0
        clrf    DDRB, F         ; port b all ouptuts
        clrf    PORTB, F        ; and all low.
        return

Init_serial initializes the USART. The baud rate is set to 9600 for both 
transmit and receive, the receiver is enabled in an asynchronous continuous 
reception mode with 8-bit words. The transmitter is enabled in an 
asynchronous, 8-bit mode.

Init_serial
        movlb   0               ; select bank 0
        movlw   .25             ; select 9600 baud
        movfp   WREG,SPBRG      ;       /
        movlw   0x90            ; set up serial pins
        movfp   WREG,RCSTA      ;       /
        clrf    TXSTA, F        ; setup transmit status
        movlb   1               ; select bank 1
        clrf    PIR, F          ; clear all interrupts
        clrf    PIE, F          ; clear all enables
        bsf     _rcie           ; enable receive interrupt
        movlb   0               ; select bank 0
        bsf     _txen           ; enable transmit
        bsf     _peie           ; enable peripheral ints
        retfie

The Service_perf routine services any interrupts received. The routine first 
disables all peripheral interrupts, then checks the receive buffer for data. If the 
receive buffer flag is set, the data is displayed on the LEDs by the 
Service_recv routine. In either case the peripheral interrupts are 
re-enabled and the routine exited.

Service_perf
        bcf     _peie           ; disable perf int
        movlb   1               ; bank 1
        btfsc   _rbfl           ; RX buffer full?
        call    Service_recv    ; yes then service
Exit_perf
        bsf     _peie           ; enable peripheral int
        retfie
;
;  This routine services the serially received info.

Service_recv
        movlb   0
        movfp   RCREG,WREG      ; load received value
        movwf   PORTB
End_recv
        clrf    INTSTA, F       ; clear all int
        bsf     _rcie           ; enable receive     
        return

The Xmit routine increments file register COUNT and outputs it to the serial 
port.  It then waits for the key to be released before returning.

Xmit
        incf    COUNT, F
        movlb   0
        movfp   COUNT,TXREG
Debounce
        movpf   PORTA,TEMP      ;
        btfss   TEMP,1          ; Wait for key release
        goto    Debounce
        return

6.6  Source Code – PIC17C42
        list p=17c42

;        TITLE   Demo board software.
;  PortB of the PIC17C42 is connected to the 8 leds and 
;  a switch (S1) is connected to the RTCC input. The software waits for 
;  S1 to be pressed at which point a counter is incremented
;  an its value is transmitted through the serial port. The RX
;  and TX lines (pins 2 and 3) must be shorted at the DB9 Connector so that 
;  the transmitted value is sent right back to the receive buffer.
;  The receive interrupt is generated and the value in the
;  receive buffer is displayed as a 8 bit value on port B.
;  The net result is that every time S1 is pressed the LEDs will
;  increment in a binary count.
;
COUNT           equ     0x20
TEMP            equ     0x19    ; Temporary storage location
;
;
      #INCLUDE   <p17C42.inc>
;
      #INCLUDE   <demo42.inc>
;
;
        org     0
        goto    Start
;        
        org     0x0010          ; vector for rtcc interrupt
Rtcc_int
        retfie                  ; return with ints enabled
;
        org     0x0020          ; vector for peripheral interrupt
Perf_int
        goto    Service_perf    ; service the interrupts
;
        org     0x0030
;
Start                           ; start main program
        movlw   0x20            ; clear ram space
        movfp   WREG,FSR0       ; use indirect addressing
Start1
        clrf    INDF0, F        ; clear ram
        incfsz  FSR0, F         ; from 0x20 to 0xff
        goto    Start1          ;       /
        call    Init_ports      ; initialize all ports
        call    Init_serial     ; initialize serial port
Loop    
        movpf   PORTA,TEMP      ; get porta

        btfss   TEMP,1          ; check for key press
        call    Xmit            ; send next value
        goto    Loop            ; check again
;
;
Init_ports
        movlb   0               ; select bank 0
        clrf    DDRB, F         ; port b all ouptuts
        clrf    PORTB, F        ; and all low.
        return
;
;  This routine initailizes the serial port. Baud rate, receive interrupt
;  software flags are initailized.
;
Init_serial
        movlb   0               ; select bank 0
        movlw   .25             ; select 9600 baud
        movfp   WREG,SPBRG      ;       /
        movlw   0x90            ; set up serial pins
        movfp   WREG,RCSTA      ;       /
        clrf    TXSTA, F        ; setup transmit status
        movlb   1               ; select bank 1
        clrf    PIR, F          ; clear all interrupts
        clrf    PIE, F          ; clear all enables
        bsf     _rcie           ; enable receive interrupt
        movlb   0               ; select bank 0
        bsf     _txen           ; enable transmit
        bsf     _peie           ; enable peripheral ints
        retfie
;
;  This routine handles all the peripheral interrupts. In this example
;  only the receive interrupt is handled.
;  check for  receive interrupts only
;
Service_perf
        bcf     _peie           ; disable perf int
        movlb   1               ; bank 1
        btfsc   _rbfl           ; RX buffer full?
        call    Service_recv    ; yes then service
Exit_perf
        bsf     _peie           ; enable peripheral int
        retfie
;
;  This routine services the serially received info.

Service_recv
        movlb   0
        movfp   RCREG,WREG      ; load received value
        movwf   PORTB

End_recv
        clrf    INTSTA, F       ; clear all int
        bsf     _rcie           ; enable receive     
        return
;
;
Xmit
        incf    COUNT, F
        movlb   0
        movfp   COUNT,TXREG
Debounce
        movpf   PORTA,TEMP      ;
        btfss   TEMP,1          ; Wait for key release
        goto    Debounce
        return
;
        END

Appendix A.   Hardware Detail
The PICDEM-1 hardware is extremely simple and is intended to illustrate the 
ease of use of several PICmicro parts. The PICDEM-1 features the following 
hardware elements:

A.1 Processor Sockets
Although three sockets are provided, only one device may be used at a time.

• 18-pin socket
• 28-pin socket
• 40-pin socket

A.2 Display
Eight red LEDs are connected to PORTB of each processor type. The PORTB 
pins are set high to light the LEDs.

A.3 Power Supply
There are two ways to supply power to PICDEM-1:

• A 9V, 100 mA unregulated AC or DC supply can be plugged into J2.

• A +5V, 100 mA regulated DC supply can be connected to the hooks 
provided. 

PICMASTER users have a regulated +5V power supply available in the logic 
probe connector and can easily connect to the hooks on PICDEM-1 (Red 
probe to +5V and Black probe to GND). 

A.4 RS-232 Serial Port
An RS-232 level shifting IC has been provided with all necessary hardware to 
support connection of an RS-232 host through the DB9 connector. The port is 
configured as DCE, and can be connected to a PC using a straight through 
cable.

PIC17C4X devices have their RX and TX pins tied to the RX and TX lines of 
the Max232A. The other devices do not have a serial port implemented in 
hardware, but an RS-232 interface may be implemented in software using 
pins RA2 and RA3.

Note: The exact power up sequence listed in Getting Started (Section 2)
of this guide must be followed.

A.5 Switches
Three switches provide the following functions:

• S1 – Real-Time Clock/Counter (RTCC), or Timer0, input

• S2 – MCLR to hard reset the processor

• S3 – Active low switch connected to RA1

Switches S1 and S2 have debounce capacitors whereas S3 does not, 
allowing the user to investigate key debounce techniques.

When pressed, the switches are grounded. When idle, they are pulled high 
(+5V).

A.6 Oscillator Options
• RC oscillator (2 MHz approximately) supplied. This oscillator may be 

disabled by removing jumper J3.

• Pads provided for user furnished crystal and two capacitors.

• Socket provided for clock oscillator.

A.7 Analog Input
A 5K ohm potentiometer is connected through a series 470 ohm resistor to 
AN0.

The pot can be adjusted from VDD to GND to provide an analog input to the 
parts with an A/D module. Parts without A/D will read a digital high or low 
depending upon the position of the potentiometer.

A.8 Sample Devices
A sample part programmed with a simple program is included.

Table A.1 lists the I/O features and port connections for each processor type.

Table: A.1  PORT CONNECTIONS

Processor LEDs
RS-232
TX/RX(1)

S1
RTCC

S2 S3
CH0 POT

R2(2)

PIC16C5X,
PIC16C7XX,
PIC16X8X

PORTB RA2/RA3 T0CKI MCLR RA1 RA0

PIC17C4X PORTB RA2/RA3 T0CKI MCLR NC(3) NC(3)

Note 1: USART module available only on PIC17C4X devices.
2: A/D module available only on PIC16C7XX devices.
3: NC: No connection

A.9 Board Layout and Schematics
The following figures show the parts layout (silkscreen) and schematics for 
the PICDEM-1 board.

Figure A.1:  PICDEM-1 Parts Layout

Figure A.2  PICDEM-1 Schematic 

